帳票設定の難形ファイルにて行える データの加工方法

Excel 上でデータを加工する一般的なテクについてご紹介致します。 御社独自の帳票の作成などにお役立て頂ければ幸いです。

目次

1.	は	じめに	2
2.	MC	その設定	2
2	.1.	別セルに出力する必要がある場合の設定方法	2
3.	Ex	cel 基礎知識	3
3.	.1.	表示形式	3
3	.2.	関数	5
3	.3.	セル参照	8
4.	汎月	用的なデータ加工方法 1	4
4	.1.	別々のセルに出力されるデータを結合する方法1	4
4	.2.	1つのセルに出力されるデータを分割表示する方法1	5
4	.3.	とあるセルの値により表示するデータを変える方法1	9
4	.4.	疑似的なチェックボックスの作成方法2	1
	上彩	吸者向け2	3
5.	特分	ミのデータ加工方法2	7
5.	.1.	郵便番号の先頭に「〒」を付ける方法2	7
索引			0

1. はじめに

MG から帳票設定やエクスポート設定、CSV 出力等でデータを出力することが可能となっています。しかしながら、出力されたデータの形式が固定であり、これを MG 側で変更する機能は現状ございません。

このマニュアルでは特に帳票設定の雛形ファイルにて効果的な方法をご紹介致します。

ここで紹介する応用術はインターネットで検索すれば出てくるような一般的なテクです。 業務上、Excelを使うことはよくあることだと思いますので、応用して頂ければ様々な面で お役に立てるかと思われます。

2. MG の設定

前提として MG からデータを出力する過程が必ず必要です。

従って、帳票設定にて出力する設定を行って頂く必要がございます。その際、加工前とな る出力データをどこに出力させるかはどの方法を取るかにより変わりますので、それぞれ の詳細をご覧下さい。

2.1. 別セルに出力する必要がある場合の設定方法

加工前と後のデータが同一のセルである場合は特に問題ないと思いますが、異なるセルに出力する必要がある場合は下図をご参照下さい。



3. Excel 基礎知識

加工に必要な知識をまずは簡単にご紹介致します。既にご存知の方は読み飛ばして下さいませ。

3.1. 表示形式

簡単に申しますと、その名の通り、見た目のデータ形式を変化させる機能です。



この機能を利用して、予め特定のデータが特定のセルに入るのを見越して形式を設定、その Excel を帳票設定の雛形ファイルとして設定して置くことでダウンロードした際に見た 目を手修正する必要性がなくなると言った使い方がございます。

この「表示形式」以外にも見た目の形式を変えることはできますが、これが一番簡単な方 法な為、表示を変えたいと思った際にはまずはこの機能を検討すると良いと存じます。

また、①をクリックした後に出てきたリスト以外の形式にすることも可能です。設定手順 は下図の通りとなりますが、⑦にて設定出来る書式は様々なものがございますので、現時点 ではご紹介は控えさせて頂きます。まずは手順をご承知いただければ幸いです。



3.2. 関数

※少し長いのですが、様々な基礎要素をご紹介しておりますので、最後までお読み頂ければ 幸いです。

B2	Ŧ	: X	🗸 fx	あああ		
- 1	А	В	С	D	E	F
1						
2		あああ				
<u> </u>						

文字を記載すると、右上(赤線で囲った部分)に同じ文字が表示されることはご存知かと 思います。ここには名前があり、「数式バー」と言います。

その名の通り、数式を書けば自動的に計算してくれます。

B2	B2 ▼ : × √ fx =1+2							
	A	В	С	D	E			
1								
2		3						
3								

しかしながら、数値が変わる度に数式を書き換えていたら手間です。その為に「1」や「2」 と言った数値は別のセルに記載し、その値を引っ張ってきて足し算させることにしましょ う。



意味:D2 セルの数値とD3 セルの数値を足す

これだけでは余計に難しくなったように見えます。直接数値を書くのではなく、数値を書 いたセルを参照する利点は数値が変化した時にあります。

では実際に変えてみましょう。

B2	*	: 🗙	✓ f _x	=D2+D3
1	А	В	С	D
2		7		3
3 4				4

こちらが実際に数値を変えた様子です。

本来であれば=3+4 と数式の記述を変えるべきなのですが、数式に実際の数値は書いていません。よって、数式ではなく、数式の中で参照しているセルに書かれた数値を変えることで計算内容を変更しました。

つまり、そのセルの値が変わろうとも数式を変更する必要はないのです。これが数値を書 いたセルを参照して数式を書く利点です。



セル参照方法とその利点をご紹介したところで、本題である関数のご紹介を致します。 関数というものは数多くございますが、ここでは一番メジャーであり、かつ分かりやすい SUM 関数をご紹介しつつ、関数についてお話致します。

では早速、関数を使った様子をご紹介致します。

SU	SUBSTIT ▼ : 🗙 🖌 ƒx =SUM(D2,D3)									
	A	В	С	D	E					
1										
2		=SUM(D2,0		3						
3				4						
4										

数式を記述している様子

B2	-	\pm X	🗸 fx	=SUM(D2	2,D3)
	А	В	С	D	E
1		7		3	
3 4				4	

数式が計算された結果

数式:=SUM(D2,D3)

意味:D2 セルの数値とD3 セルの数値を足す

これでは先程同様、全く利点が伝わっていないと存じます。

何の関数であるかによって利点は変わってきますが、本来なら手動でやらなければなら ないことを一定のルールで記述をしてあげれば簡単に行えるというのは共通する利点だと 思われます。何よりもセル参照と組み合わせることでその利点がより活かせますので、デー タを増やして SUM 関数固有の利点と共にセルの範囲指定についてもご紹介致します。

SU	BSTIT 🔻	: 🗙	🖌 fx	=SUM(D2	2:D9)
	A	В	С	D	E
1					
2		=SUM(D2:0		3	
3				4	
4				5	
5				6	
6				7	
7				8	
8				9	
9				10	

数式を記述している様子

B2	-	: ×	$\checkmark f_x$	=SUM(D2	2:D9)
	А	В	С	D	Е
1					
2		52		3	
3				4	
4				5	
5				6	
6				7	
7				8	
8				9	
9				10	

数式が計算された結果

数式:=SUM(D2:D9)

意味: D2 セルから D9 セルまでの数値を全て足す

如何でしょうか。

直接数値を書くような場合、3から10を足すという数式は書けないかと思います。例え 関数を使っても=SUM(3,4,5,6,7,8,9,10)と長々と数値の数だけ数式を長く書く必要があり ます。

では逆にセル参照だけで関数を使わない場合の記述方法は=D2+D3+D4+D5+D6+D7+ D8+D9 と書く必要があります。つまり、数値が書かれたセル分だけこちらも数式が長くな るのです。

そこで SUM 関数の特性を利用します。SUM 関数は単に足し算をしてくれるというもの ではなく、数値さえ与えてあげればそれを全て足してくれるという関数です。従って、セル の範囲指定という機能を使用して、D2 セルから D9 セルまでという範囲指定をします。こ れにより、D2 セルから D9 セルまでの数値を SUM 関数に与えたことになります。このよ うにして D2 セルから D9 セルまでの数値を全て足してもらうという状況を作り上げること が出来るのです。

この方法ですと数値が増えてもセルの範囲を変える必要はありますが、数式が長くなることはなくなります。

	きとめ	
•	セル参照の中に範囲排	旨定という方法がある
	記述ルール: {開始セ	ル?:{終了セル}
•	関数は手動ですると手	F間なことを、記述ルールを守って記載すれば
	簡潔に行うことができ	きる
•	セル参照と関数を併用	目することにより、互いの利点をより引き出す
	ことが可能	

3.3. セル参照

セルの参照は関数においては必須の手段です。「<u>3.2. 関数</u>」にて既にご紹介しているもの もありますが、改めて順を追ってご紹介させて頂きます。

B2		Ŧ		\times	\checkmark	fx.	=1+2
	A	В		С		D	
1							
2			3				
3							

まず、セル参照というものは参照したセルの値を参照するというものです。

例えば、上図のように B2 セルに「1」と「2」を足すというだけの数式を書いたとします。 当然答えは「3」です。ではこのセルをセル参照で別のセルに表示してみます。

02		Ŧ	:	\times	\checkmark	fx.	=B2
1	A	В		С		D	
2			З		3		

上図の通り、=1+2 という数式が書かれている B2 セルを C2 セルがセル参照しています。 しかし、表示されているのは「3」です。つまり、セル参照によって参照しているのは値で あり、数式ではありません。これは結構な違いではあるのですが、基本は意識しないでも問 題ないと思われます。

では、セル参照の機能についてご紹介したところで、セル参照の種類についてご紹介して いきたいと思います。数式を書いている途中で別のセルをクリックしたらセル参照はでき てしまうのですが、それだけでは不都合が起きることがありますのでご紹介致します。

	Α	В	С	D	E
1					
2		商品名	単価(円)	金額(円)	←計算式
3		商品A	2,180	2,354	=C3*1.08
4		商品B	1,682	1,817	=C4*1.08
5		商品C	3,380	3,650	=05*1.08
6		商品D	2,480	2,678	=C6*1.08
7		商品E	1,880	2,030	=07*1.08
-					

今回は複数の関数を同時にご確認頂いた方が分かりやすい為、右隣に計算式を記述させ て頂いております。上図は C 列に記載された単価に消費税を掛けて金額を出しているとい うものです。D 列に記載した計算式は E 列に記載しています。

これで通常問題はないのですが、消費税は時代によって異なります。では消費税が変わった時にどうするかと言うと、下図のように一つ一つ数式を修正することとなります。

	А	В	С	D	E
1					
2		商品名	単価(円)	金額(円)	←計算式
3		商品A	2,180	2,398	=03*1.10
4		商品B	1,682	1,850	=C4*1.10
5		商品C	3,380	3,718	=05*1.10
6		商品D	2,480	2,728	=06*1.10
7		商品E	1,880	2,068	=07*1.10
-					

実際にしてみなくてもこれがとても面倒くさいことは分かると思います。面倒なだけな らまだしも、数式の書き間違えが発生したりして結果が異なってくると困ります。従って、 セル参照を利用して消費税も別セルに書いておいて参照する形にすることにしてみましょ う。

	А	В	С	D	E	F
1						0.08
2		商品名	単価(円)	金額(円)	←計算式	
3		商品A	2,180	2,354	=C3*(1+F1)	
4		商品B	1,682	1,850	=04*1.10	
5		商品C	3,380	3,718	=05*1.10	
6		商品D	2,480	2,728	=06*1.10	
7		商品E	1,880	2,068	=07*1.10	
0						

F1 セルに消費税「0.08」を記載してみて、D3 セルのみ書き換えてみました。ここで先程 同様に D4 セルから D7 セルまで数式を一つ一つ修正していってもいいのですが、先程も記 載したように面倒なだけならまだしも、数式の書き間違えが発生したりして結果が異なっ てくると困ります。ということで D3 セルをコピーして D4 セルに貼り付けてみましょう。

	А	В	С	D	E	F
1						0.08
2		商品名	単価(円)	金額(円)	←計算式	
3		商品A	2,180	2,354	=C3*(1+F1)	
4		商品B	1,682	1,682	=C4*(1+F2)	
5		商品C	3,380	3,718	=05*1.10	
6		商品D	2,480	2,728	=06*1.10	
7		商品E	1,880	2,068	=07*1.10	
0						

上図の通り、参照したセルが自動的にそのセルに合わせて修正されています。C3 セルが C4 セルに、F1 セルが F2 セルになっています。

ここでコピーして貼り付けをした際に何が起こったのかを簡単に解説させて頂きます。 まず、D3 セルをコピーした時には何も起こっておらず、D4 セルに貼り付けを行った際に Excel が自動的に修正を行うということが行われているのです。具体的には D3 セルから何 列何行ずれたセルに貼り付けを行ったのかを Excel が検知して、数式の中の参照されたセ ルをそのずらされた分だけずらすということをするのです。従って、そのまま D5 セルに貼 り付けを行うと下図のようになります。

	А	В	С	D	E	F
1						0.08
2		商品名	単価(円)	金額(円)	←計算式	
3		商品A	2,180	2,354	=C3*(1+F1)	
4		商品B	1,682	1,682	=C4*(1+F2)	
5		商品C	3,380	3,380	=C5*(1+F3)	
6		商品D	2,480	2,728	=06*1.10	
7		商品E	1,880	2,068	=07*1.10	
0						

D3 セルから D5 セルは 0 列 2 行下にずれがあります。従って C3 セルが 0 列 2 行下にず らされ C5 セルに、F1 セルが 0 列 2 行下にずらされ F3 セルになっているということです。 さて、同様に D6 セル、D7 セルにも貼り付けを行ったところで計算結果を見てみましょう。

	А	В	С	D	E	F
1						0.08
2		商品名	単価(円)	金額(円)	←計算式	
3		商品A	2,180	2,354	=C3*(1+F1)	
4		商品B	1,682	1,682	=C4*(1+F2)	
5		商品C	3,380	3,380	=C5*(1+F3)	
6		商品D	2,480	2,480	=C6*(1+F4)	
7		商品E	1,880	1,880	=07*(1+F5)	
~						

上図の通り、D3 セル以外は正しい計算結果になっていません。それもそのはずで、D4 セルから D7 セルは空欄の F2 セルから F5 セルを参照した数式となっているからです。本来、 ここは F1 セルを参照するべきであり、F1 セルを参照さえしていれば計算結果も問題ない はずです。

ここでするべきなのは D3 セルをコピーし、貼り付けた際に F1 セルから動かないように するというものです。では、まず D3 セルにカーソルを合わせ、更に D3 セルの数式の中の 「F1」と言うセル参照部分を部分をクリックして下さい。そのままキーボードの F4 キーを 押してみて下さい。すると「F1」と言う部分が「\$F\$1」になったかと思います。Enter キ ーを押して数式を確定した後に計算結果を見てみると、特に変化はしていません。何が変わ ったのかは先程同様、D3 セルをコピーして、D4 セルに貼り付けてみると分かりやすいで す。では実際してみましょう。

	А	В	С	D	E	F	
1						0.08	
2		商品名	単価(円)	金額(円)	←計算式		
3		商品A	2,180	2,354	=C3*(1+\$F\$1)		
4		商品B	1,682	1,817	=C4*(1+\$F\$1)		
5		商品C	3,380	3,380	=C5*(1+F3)		
6		商品D	2,480	2,480	=C6*(1+F4)		
7		商品E	1,880	1,880	=C7*(1+F5)		
_							

先程と違い、F1 セルを参照したままになっているのが分かると思います。しかし、先程のまま何も変えていない「C3」というセル参照部分は先程同様「C4」に自動的に変わっています。つまり、「F1」とセルの番地を書いただけでは自動的に変えられてしまうけれども、

「\$F\$1」というセル参照の仕方をするだけで自動的に修正されなくなるのが「\$」です。つまり、「\$」という文字が固定を表していると思って頂ければ分かりやすいかと思います。

では、このまま D5 セルから D7 セルにも貼り付けて、D5 セルから D7 セルへの数式も 問題がないこと及び計算結果が正しいことを確認してみましょう。

	А	В	С	D	E	F
1						0.08
2		商品名	単価(円)	金額(円)	←計算式	
3		商品A	2,180	2,354	=C3*(1+\$F\$1)	
4		商品B	1,682	1,817	=C4*(1+\$F\$1)	
5		商品C	3,380	3,650	=C5*(1+\$F\$1)	
6		商品D	2,480	2,678	=C6*(1+\$F\$1)	
7		商品E	1,880	2,030	=C7*(1+\$F\$1)	
0						

上図の通り、問題なくできました。では固定にしたいセル参照の個所を F4 キーで「\$」 を付ければいいのかというと確かにその通りなのですが、話はもう少し続きます。

「\$」は先程申しました通り、固定を表すものだと思って頂いて構いませんが、ここで考えるべきなのは何を固定しなければならないのかというところです。話をする前に実際にしてみた方が分かりやすいかと思いますので、してみましょう。

初めに先程と同様に D3 セルの数式の中の「\$F\$1」と言うセル参照部分を部分をクリックして、キーボードの F4 キーを押してみて下さい。すると「F\$1」に変わったかと思います。Enter キーを押して確定した後、D3 セルを D4 セルから D7 セルまで貼り付けてみて下さい。

	А	В	С	D	E	F
1						0.08
2		商品名	単価(円)	金額(円)	←計算式	
3		商品A	2,180	2,354	=C3*(1+F\$1)	
4		商品B	1,682	1,817	=C4*(1+F\$1)	
5		商品C	3,380	3,650	=C5*(1+F\$1)	
6		商品D	2,480	2,678	=C6*(1+F\$1)	
7		商品E	1,880	2,030	=C7*(1+F\$1)	
0						

上図の通り、F1 セルを参照しているという部分は固定のまま貼り付けられました。F1 セルを参照しているのだから計算結果も勿論変わりません。

続いて、もう一度 D3 セルの数式の中の「F\$1」と言うセル参照部分を部分をクリックして、キーボードの F4 キーを押してみて下さい。すると今度は「\$F1」に変わったかと思います。Enter キーを押して確定した後、D3 セルを D4 セルから D7 セルまで貼り付けてみて下さい。

	А	В	С	D	E	F
1						0.08
2		商品名	単価(円)	金額(円)	←計算式	
3		商品A	2,180	2,354	=C3*(1+\$F1)	
4		商品B	1,682	1,682	=C4*(1+\$F2)	
5		商品C	3,380	3,380	=C5*(1+\$F3)	
6		商品D	2,480	2,480	=C6*(1+\$F4)	
7		商品E	1,880	1,880	=C7*(1+\$F5)	
-						

先程とは異なり、コピーしたことで F1 セルを参照しているという部分が固定されずに D3 セルからずらした分だけ F1 セルの参照もずれてしまっています。当然、計算結果もそ れに伴って変わってしまっています。

セル参照の種類はこれで以上です。続いて D3 セルの数式の中の「\$F1」と言うセル参照 部分を部分をクリックして、キーボードの F4 キーを押してみても「F1」という最初に記載 したセル参照方法に戻るだけです。

ここで解説をさせて頂きますと、そもそも「F1」というセル番号は「F」という列と「1」 という行で成り立っています。その前に固定の印である「\$」を付けることで列を固定する のか行を固定するのかを指示することが出来るのです。

D4 セルは D3 セルより 0 列 1 行下にずれていますので、D3 セルを D4 セルに貼り付け るならば初めから自動修正される心配のない列を固定する必要はありません。逆に行は自 動修正される心配がある為、固定する必要があります。よって行の前にのみ「\$」を付け、

「F\$1」とすれば良いという結論に至ります。

勿論、「\$F\$1」にしておいても構わないのですが、列をずらして貼り付けをすることがないのであれば特に意味のない固定ということになります。どちらの方法を取るにしろ、この違いは今後の為に必要となるものなので覚えておいて頂けますと幸いです。

4. 汎用的なデータ加工方法

雛形ファイル、つまり Excel 上でデータを加工する方法についてご紹介致します。 尚、目的に対して手段は一つではないですし、紹介した方法を基盤として応用することも 可能なはずなので、どうしたいかを明確にした上でご覧頂けましたら幸いです。

4.1. 別々のセルに出力されるデータを結合する方法

	₽	終目標		
- 1	A	В	С	D
1				
2	住所	東京都港区西新橋1-13-1		
3		1		東京都
4				港区西新橋1-13-1
5		このようにそれぞれのセル		1
6		のデータを結合して表示す		
7		ることが可能です。	マカス	
8				
9			6.07	こナーダと致します。
10				
4.4				

これを実現している数式は非常に簡単です。下図をご覧下さい。

su	SUBSTIT • : \times \checkmark f_x =D3&D4					
	А	В	С	D		
1						
2	住所	=D3&D4				
З				東京都		
4				港区西新橋1-13-1		
5						

数式: =D3&D4

意味: D3 セルのデータと D4 セルのデータを結合して表示

以上です。つまり、それぞれのセルにある文字列を結合するには「&」でセルを繋ぐだけ となります。勿論、下図の通り「&」を使って結合するのに数制限はありません。

*	: 🗙 🗸 🏂 =D3&D4&D5		
A	В	С	D
住所	東京都港区西新橋1-13-1 DLXビルディング 6F		
			東京都
			港区西新橋1-13-1
			DLXビルディング 6F
	→ A 住所	 ▼ : × √ fx =D3&D4&D5 A B 住所 東京都港区西新橋1-13-1 DLXビルディング 6F 	 ▼ : × ✓ fx =D3&D4&D5 A B C 住所 東京都港区西新橋1-13-1 DLXビルディング 6F

数式: =D3&D4&D5

意味: D3 セルのデータと D4 セルのデータと D5 セルのデータを結合して表示

また、「&」で結合できるのはセル参照だけではなく、直接文字列を記載することも可能 です。例えば、D4 セルと D5 セルのデータの間にスペースを入れたいという場合もあると 思います。そのような場合は下図のように致します。

B2	-	: 🔀 🗸 ƒ =D3&D4&" "&D5		
	A	В	С	D
1				
2	住所	東京都港区西新橋1-13-1 DLXビルディング 6F		
3				東京都
4				港区西新橋1-13-1
5				DLXビルディング 6F
6				

数式:=D3&D4&" %D5

意味:D3 セルのデータとD4 セルのデータと「」とD5 セルのデータを結合して 表示

数式の中に直接記述する文字列は「"」で括ってある必要があります。今回はスペースを 間に入れる為にスペースを「"」で括って、それを「&」で結合しました。

このように固定の文字列は直接記述を使って記述し、可変の文字列はセル参照を行うことをおススメ致します。

4.2. 1つのセルに出力されるデータを分割表示する方法

	最終日	3 標				
	A	В	С	D	E	F
1						
2		田中	太郎		田中 太郎	
3					1	
4				_		
5		、一人の前2	1後とじこのま		MGより出力し	た本来
6	210	- 別でルに分	`剖して衣示の	9	のデータとしる	ます。
7		ることか可	ন্ট্ প্রি			
0				_		

もしこのようなことを行いたい場合、少々工夫が必要となってまいります。

まず第一に明確な**区切り文字**が必要です。上図で言うとスペースが区切り文字に当たり ます。これはどの文字を基準にして別々のセルに分けるのかというルールを設定するのに 必要となります。

上図では B2 セルに区切り文字であるスペースの前までの文字を表示させています。そして C2 セルには区切り文字であるスペースの後からの文字を表示させています。

このような場合、数式はそれぞれ異なります。まずは B2 セルの区切り文字の前までを表示させる方法をご覧下さい。

SU	BSTIT 🔻	: 🗙	$\checkmark f_x$	=LEFT(E2	2,FIND(" ",E2)-	1)
	A	В	С	D	E	F
1						
2		=LEFT(E2,F	太郎		田中 太郎	
3						

数式:=LEFT(E2,FIND(" ",E2)-1) 意味:E2セルを先頭(左端の文字)から""の一文字前の文字までを抽出する 展開:=LEFT(E2,FIND(" ",E2)-1) =LEFT(E2,FIND(" ","田中 太郎")-1) =LEFT(E2, 3-1) =LEFT("田中 太郎",2) ="田中"

2つの関数を使っていますので、少々分かり辛いかもしれません。しかし、関数を組み合わせることにより様々な機能が実現できるというのが関数の奥深くも魅力的な性質だと思います。

今回は LEFT 関数と FIND 関数と言う以下のような関数を使いました。



今回、区切り文字が必要と言うのは文字列の抽出を行う為には何文字目までという指定 が必要なのにも関わらず、何文字目までを抽出するかはデータ次第だからです。

例で述べるのならば、人の姓が必ず 2 文字と決まっているのであれば=LEFT(E2,2)で問 題はありません。しかし、現実的に考えて姓を 2 文字であると仮定するのは難しいでしょ う。従って、特定の文字の場所を検索する FIND 関数と組み合わせることでこの問題を解 決しております。

では実際に想定されるケースを見てみましょう。

B2	A	• : 🗙 B	<i>✓ fx</i> ⊂	=LEFT(E	2,2) <- 「田中」という 2 文字の姓 なので問題ありません。
1		田中	太郎		田中太郎
B2	A	• : X B	✓ ƒx c	=LEFT(E2	2,2) < 「五十嵐」という3文字の姓 だと想定通りになりません。
1 2 3		五十	次朗		五十嵐次朗
B2		r = 🗙	$\checkmark f_x$	=LEFT(E2	,FIND("",E2)-1)
1	A	В	С	D	E F
1 2 2	A	B 五十嵐	C 次朗	D	E F 五十嵐 次朗
1 2 3	A 区切り文字 姓の文字数 想定通り	B 五十嵐 を使う方法た なが変わっても となります。	O 次朗 と	D	E F 五十嵐、次朗 但し、データに区切り文字が 存在することが前提です。
1 2 3	A 区切り文字 姓の文字数 想定通り	B 五十嵐 こを使う方法た なが変わっても となります。	C 次朗 と ノノ 人	D =LEFT(E2	E F 五十嵐、次朗 但し、データに区切り文字が 存在することが前提です。 2,FIND("",E2)-1)

如何でしょうか。

データに区切り文字が存在すること前提ではありますが、この数式ならば姓が何文字あ ろうとも想定通りに分割出来ることが伝わったかと思います。

実際にはデータの記述ルール含め、色々考慮しつつ柔軟に関数の組み合わせを行う必要 がありますが、今回の例ではこれを最適解と致します。

続いて、C2 セルの区切り文字の後からを表示させる方法をご覧下さい。

SU	BSTIT 🔻	: 🗙	🗸 fx	=RIGHT(2,LEN(E2)-FIN	D(" ",E2))
	A	В	С	D	E	F
1						
2		田中	=RIGHT(E2,		田中 太郎	
3						



こちらは3つの関数を併用しております。FIND 関数は先程も使用した為、ご紹介は省略 させて頂きます。

RIGHT 関数	
=RIGHT(<i>文字列</i> , <i>文字数</i>)	
<i>文字列</i> を右端から <i>文字数</i> 分抽出する	
例)=RIGHT("あいうえお",3)	
計算結果:うえお	
LEN 関数	
LEN 関数 =LEN(<i>文字列</i>)	
LEN 関数 =LEN(<i>文字列</i>) <i>文字列</i> が何文字かを返す	
LEN 関数 =LEN(<i>文字列</i>) <i>文字列</i> が何文字かを返す 例) =LEN("あいうえお")	

今回は末尾(右端)から抽出する文字数を指定する形です。しかし、FIND 関数は先頭(左端)から何文字かという答えしか返してくれません。よって、全体で何文字ある文字列であるかを LEN 関数で計算させた後にその差分を抽出文字数として RIGHT 関数に指定するという一手間掛けた方法で実現しています。

しかしながら基本的な方法はどちらも同じです。文字数が固定である場合はそんなに悩 む必要はないと思いますが、可変である場合は以上を参考に設定して頂けると幸いです。

	最終	目標				
	A	В	С	D	Е	
1						
2		30代		34		
3		1		1		
4						
5	یا ا	リように本米		MGより出ナ	」した本来	
6		ータから別		のデータと	:します。	
7	テー	-외こ加工す	ର <u>୍</u> ଗ୍			
8	2	が出来ます	•			
9						

4.3. とあるセルの値により表示するデータを変える方法

こちらは内容的には難しいことは何も行っていません。ただパターンの数だけ関数を使 う為に式が長く、難しく見えるかもしれません。しかし先に述べた通り、内容は非常にシン プルです。

▼ : 🗙	✓ fx	=IF(D2>= IF(D2>=3	=IF(D2>=60,″60歳以上″,IF(D2>=50,″50代″,IF(D2>=40,″40代″, IF(D2>=30,″30代″,IF(D2>=20,″20代″,″19歳以下″)))))											
В	С	D	E	F	G	Н	Ι							
[下")))))		34												

数式:=IF(D2>=60,"60歳以上",IF(D2>=50,"50代",IF(D2>=40,"40代
",IF(D2>=30,"30代",IF(D2>=20,"20代","19歳以下")))))
意味:E2セルの値が60以上なら「60歳以上」と表示、
E2セルの値が50以上なら「50代」と表示、
E2セルの値が40以上なら「40代」と表示、
E2セルの値が30以上なら「30代」と表示、
E2セルの値が20以上なら「20代」と表示、
E2セルの値がそれ以外なら「19歳以下」と表示

長い数式ではありますが、上述したように内容はシンプルです。この式を展開する前にこの数式についてお話したいと思います。

まず使用している関数は IF 関数のみです。そして次に同じ関数を組み合わせて一つの数 式を作り上げています。複数の関数を組み合わせるということは前章でも行いましたが、同 じ関数でもこれは可能です。つまり関数の使用にも回数制限はないということです。

具体的に IF 関数とはどういう関数なのかというと下記の通りとなります。

IF 関数
=IF(<i>論理式,TRUE の時の返り値,FLASE の時の返り値</i>)
<i>論理式</i> の真偽による返り値を返す
例)=IF(34>30,"大きい","小さい")
計算結果:大きい

論理式と言うより条件と言った方が分かりやすいかもしれません。対象と何かを比較し、 その結果により行動が分岐するという受動的な行動が生じます。

今回の数式のように文字列を返すだけでなく、論理式の結果を受けて更に論理式を計算 させることも可能です。これにより一つ一つの論理式を簡潔にすることが可能です。

それでは実際に今回の数式の展開する様子を見てみましょう。

今回は数式が長く、該当箇所が分かり辛いので見やすいように一部を「***」と表現致します。
展開:=IF(D2>=60,"60 歳以上",***)
=IF(34>=60,"60 歳以上",IF(D2>=50,"50 代",***))
=IF(FALSE,"60 歳以上",IF(34>=50,"50 代",IF(D2>=40,"40 代",***)))
=IF(FALSE,"60 歳以上",IF(FALSE, "50 代",IF(D2>=40,"40 代",IF(D2>=30,"30 代 ",***))))
=IF(FALSE,"60 歳以上",IF(FALSE, "50 代",IF(FALSE,"40 代",IF(34>=30,"30 代 ",IF(D2>=20,"20 代","19 歳以下")))))
=IF(FALSE,"60 歳以上",IF(FALSE, "50 代",IF(FALSE,"40 代",IF(TURE,"30 代 ",IF(34>=20,"20 代","19 歳以下")))))
=IF(FALSE,"60 歳以上",IF(FALSE, "50 代",IF(FALSE,"40 代",IF(TURE,"30 代 ",IF(FALSE,"60 歳以上",IF(FALSE,"50 代",IF(FALSE,"40 代",IS(14>=20,"20 代","19 歳以下")))))
=IF(FALSE,"60 歳以上",IF(FALSE, "50 代",IF(FALSE,"40 代","30 代")))
=IF(FALSE,"60 歳以上",IF(FALSE,"50 代","30 代"))
=IF(FALSE,"60 歳以上",IF(FALSE,"50 代","30 代"))

関数は基本左側から実施されますので、IF 関数も例に漏れず一番左の IF 関数にあるセル 参照の値を展開し、論理式の計算を行います。その真偽の判定結果によりどちらかを実施し ます。今回の数式は真(TRUE)になった時点で出力値が決定する構成ですので、真(TRUE) になるまで繰り返し IF 関数の論理式を計算するという流れになっています。

IF 関数は何かしらの受動的な行動分岐を起こしたい場合の基本的な関数になるので、是 非ご活用頂ければ幸いです。 4.4. 疑似的なチェックボックスの作成方法

	ł	ł			B	ł	Ŗ																						
	А	В	С	D	Е	F	G	Н	I	JI	K L	. M	N	0	P	Q	R	S	Т	U	V	W		Х			Y	Z	
1																							選択選択	えたう えたま	t ≢				
2			選:	択肢	壱		Ø	選打	尺肢	弐] 選	択朋	技参	-														
3		\square	選:	択肢	肆			選打	尺肢	伍] 選	択肢	技陸										14:	x1t	zιλ			
4																									自由]項	目チェッ	ックボック	
5					5	M	B2	62	12 6	22 G	313	tz 11	,								_			_	スの	デ・	ータをこ	こに出力	
6						1.	疑	仙角	, LZ, L ()な=	,с, Ғт,	いわ7	tin.	יבר	を作	চিন	, ŧI	,t-	_			_			_	した	と仮	反定しま	す。	
7							X1	-17川	, ທຸ	カ容	ຫຼື	ÞI 22	右に	ある	5選	択膨	おが	。 存在	Ð	H	_				_				_
8						-	t=	場合	、方	≓0-1	f _I	ックオ	Б. Г.	クス	にチ	- - - -	~ ,ウt	が付	<	H	-								
9						-	દા	13	動作	=	 可能	にな	るよ	31	207	ເເ	ます		`	H	_				_				
11																													
10				А	В	С	D	Е	F	G	Н	Ι	J	К	L	Μ	Ν	Ο	Ρ	Q	R	: 5	3 T	U	V	W	\rightarrow	(
_		1																									選択規	技壱	
																											選択	技参	
			1																								選択規	技肆	
			2		\square	選	択朋	伎壱	5		選	択肢	戝			選:	択肢	技参											
			3		\square	選	択刖	伎肆	t		選	択肢	伍			選	択肢	姬			当	然	X1	の内	容7	が変	_โ อก	£	
			4																		白	勈	र्भग व	Ťτ	w/t	thic		5	
			5																		t-1	10)EEL	12-	t-1	り致	ਹ=ਹ ਸ਼,≢ ਰ		
																						-							
			6																							,		<u> </u>	

自由項目チェックボックスを出力すると、チェックを付けた選択肢が改行区切りで1セルに纏まって出力されます。その中に選択肢が存在していたら、上図のように予め用意して置いた空のチェックボックス(□)にチェックが付いた状態のチェックボックス(☑)になるというものです。

※マッチングッドに作っている選択肢とこの選択肢が同じこと前提です。

実際どのような関数を使っているかをご紹介します。

SUBSTIT 🝷 : 🗙									$\checkmark f_x$ =if(counting)									F(\$X\$1, "*選択肢壱*")," ☑"," □")						
	А	В	С	D	Е	F	G	Н	Ι	J	К	L	М	N	0	Ρ	Q	R	S	Т	U	V	W	X
1																								選択肢弐 選択肢肆
2		=IF	選:	択肋	壱			選	択肢	戝			選	択肋	錂									
3		Ø	選:	択肋	肆			選	択肢	友 伍			選	択肋	姬									

数式:=IF(COUNTIF(\$X\$1."*選択肢壱*")."☑"."□")	
意味:X1セルに「選択肢壱」が含まれていたら「☑」、記載されていなけ	ければ
「□」と表示する	
展開: <mark>=IF(COUNTIF(\$X\$1, "*選択肢壱*"),"☑","□")</mark>	
=IF(COUNTIF("選択肢弐	
選択肢肆", "*選択肢壱*"),"☑","□")	
=IF(0,"⊠","□")	
="□"	

このように以前に紹介した IF 関数と COUNTIF 関数を組み合わせるだけで疑似的なチェックボックスの作成が出来ます。

尚、Excel において「0」は「FALSE」つまり「偽」と同値であり、「0」以外は「TRUE」 つまり「真」と同値として扱われます。

それでは COUNTIF 関数が何をしている関数であるかをご紹介致します。

COUNTIF 関数	
=COUNTIF(範囲,檢索条件)	
指定された <i>範囲</i> 内のセルの中に指定した検索条件に合うセルが幾つ存	
在するかを返す	
例)=COUNT(A1:A3, "あいう")	
前提:A1「あいう」、A2「えお」、A3「あいう」	
計算結果:2	

本来、COUNTIF 関数は上述した通り、セルをカウントする為の者です。しかし、今回は 参照した検索条件に合うセルであるかどうかを判断するという部分を利用して、参照した セルの中に選択肢が存在するか否かを判断させています。COUNTIF(X1, "*選択肢壱*")と いう部分がこれに当たります。

ここでご紹介するべきなのは「選択肢壱」という選択肢の文言の左右にある「*」という 文字です。ではまず、この「*」がなかった場合どうなるかをご紹介致します。

: 🗙	$\sqrt{f_x}$ =IF	(COUNTIF(\$X\$1, ″選	択肢壱")″☑″,″□″)
EFG	H I J K L	MNOPQRS	stuvw x
			選択肢壱
肢壱 🛛	選択肢弐 🛛 🗆	選択肢参	
肢肆 □	選択肢伍 🛛	選択肢陸	
- ट र र	r : × D E F G R肢壱 ロ : R肢肆 ロ :	r :	r : ★ ★ ƒ★ =IF(COUNTIF(\$X\$1, "選 D E F G H I J K L M N O P Q R S R肢壱 □ 選択肢弐 □ 選択肢参 R肢肆 □ 選択肢伍 □ 選択肢陸

チェックが付きます。

.

В2				•	:	: 🔀 🗸 fx =IF(COUNTIF(\$X\$1, "選択肢壱"),"☑","□")																		
	ABCDEFGHIJKI			L	LMNOPC				Q	RSTU			U	V	x									
1																								選択肢壱 選択肢弐
2			選	択肢	t壱			選択肢弐					選	択肢参										
3			選	択肢	支肆			選	沢肋	δ伍			選	沢肋	姬									
																ŗ		 3	見、 よう	ん 検 う に クカ	- 索 見 え が付	く 件 に ま ・ ま ・ ま ・ ま	 に一 すが ミセノ	致してい 、チェッ ん。

何故チェックが付かないかと申しますと、COUNTIF 関数は参照した検索条件に合うセルであるかどうかを判断します。つまり具体的には、検索条件はX1 セルの「選択肢壱 選択肢弐」という文字列と「選択肢壱」がイコールであるか否かを判定し、イコールでないという結論が出され、チェックが付かないという事態になりました。

想定通りの動きをしてもらう為には X1 セルの「選択肢壱 選択肢弐」という文字列の中に「選択肢壱」は含まれているか否かを判定してもらう必要が あります。そこで「*」を使用します。「*」とは 0 文字以上の文字列の代替記号のようなも のです。つまり、何らかの文字列があってもなくても「*」と一致する為、「選択肢壱」が含 まれているか否かを判定する為の条件に相応しいものとなります。具体的に言いますと、X1 セルが「選択肢壱」であれ「あああ選択肢壱あああ」であれ「あああ選択肢壱」であれ「選 択肢壱あああ」であれ「*選択肢壱*」とイコールということです。但し「あああ」と「*選 択肢壱*」はイコールにはなりません。「*選択肢壱*」には「選択肢壱」という文字列が必要 だからです。勿論、「*選択肢弐*」であれば「選択肢弐」という文字列が必要ということに なります。

上級者向け

ここで数式が複雑になってしまいますが、より実践向きな数式をご紹介致します。

				R		ŀ	Ξ																					
В	2			•	÷	2	ĸ	V	1	fx.		=IF	(00	1UC	JTIF	=(\$)	K\$1	," *'	" &I	NDI	RE	эт(AD	DRESS(RO	W(),COLUM	N()+1))&"*	"),″ ⊠ ″ ,″ ⊡	")
	A	В	С	D	Е	F	G	Н	Ι	J	К	L	М	N	0	Ρ	Q	R	S	Т	U	V	W	X	Y	Z	AA	
1																								選択肢壱 選択肢弐				
2			選	択肢	壱			選:	択肋	支弐			選:	択肋	錂													
3			選	択肘	肆			選:	択肋	支 伍			選:	択肋	姬													

見て頂いたら分かる通り、数式が長くなっています。基本は先程と変わらない為、違いを 見てみましょう。

=IF(COUNTIF(\$X\$1,"*選択肢壱*"),"☑","□")

=IF(COUNTIF(\$X\$1,"*"&INDIRECT(ADDRESS(ROW(),COLUMN()+1))&"*"),"☑","□")

上が前回の数式で下が今回の数式です。見た頂いたら分かる通り、緑文字部分の COUNTIF 関数の検索条件の部分のみが異なっています。従ってそこのみをピックアップ してご紹介させて頂きます。

まず、COUNTIF 関数の検索条件の部分のみをピックアップしたものを比較してみましょう。

"*選択肢壱*"

"*"&INDIRECT(ADDRESS(ROW(),COLUMN()+1))&"*"

異なっているのは緑文字の部分ですが、もう少し分かりやすくいじってみます。 "*"&"選択肢壱"&"*"

"*"&INDIRECT(ADDRESS(ROW(),COLUMN()+1))&"*"

つまり「&」は「<u>4.1. 別々のセルに出力されるデータを結合する方法</u>」でご紹介した通り、 文字列の結合に使用する記号です。そして、0文字以上の文字列と言う意味の「*」と選択 肢である「選択肢壱」を結合している以上、"*選択肢壱*"と"*"&"選択肢壱"&"*" は同じ意味 の数式です。

それを前提に上述した COUNTIF 関数の検索条件の違いを見てみると今回変わったのは 選択肢の部分のみということが分かるかと思います。

変えた理由を簡潔に言うと、選択肢は Excel 上のチェックボックスの右横のセルに書い ているのにわざわざ数式の中にも書くのは面倒なので、隣のセルの値を参照してしまおう というものです。分かりやすく書くと下図のようになります。

す す 支 式

B2 セルのチェックボックスは C2 セルにある選択肢のチェックボックスなので、選択肢の値を C2 セルを参照することで検索条件としても結果は同じというわけです。分かりやすく並べてみましょう。

"*"&"選択肢壱"&"*"

"*"&<mark>C2</mark>&"*"

"*"&INDIRECT(ADDRESS(ROW(),COLUMN()+1))&"*"

どうでしょうか。1列目のように直接選択肢を書くより、2列目のセル参照の方が楽だと 思いませんか。例え選択肢の値が変わってもC2セルの値を書き換えるだけで済みます。こ のように出来る限り将来何かあった際に書き換える箇所が少なくて済む数式の方がトラブ ルが起こりにくい為、考えられるものには対応しておく方が良いというのが通説です。勿論 過剰に対応すると求めている品質と性能やコストが合わなくなってしまいますので、程々 にするべきではあります。

ではどうしてセル参照を素直に行わず、このような難しい関数を羅列しているかと申し ますと、特定のセルではなく、チェックボックスのある右隣のセルという抽象的なセル参照 を行いたいが為です。勿論これが過剰な性能であると感じられる場合は以降をご覧頂く必 要はございません。前述した通り、過剰な性能を持たせる必要はない為、必要な場合のみご 使用下さいませ。

必要な方々向けに、COUNTIF 関数の検索条件の更に選択肢を指す部分である INDIRECT(ADDRESS(ROW(),COLUMN()+1))について解説させて頂きます。

まず前述した通りこちらの数式は右隣のセル――つまり、1列右及び0行ずれたセル―― を参照するという意味になっています。 ではまず使用している4つの関数の内2つをご紹介致します。

ROW 関数	
=ROW0	
行番号を返す	
COLUMN 関数	
=COLUMN()	
- 0	
列番号を返す	

つまり、ROW 関数と COLUMN 関数を使うことで現在のセルの場所を求めることが出 来ます。この2つはこういう関数だと覚えて頂けると幸いです。

右隣のセル――つまり、1列右及び0行ずれたセル――を参照したいので、COLUMN関数で求めた現在の列に+1することで右隣のセルの場所を求めることが出来るようになります。

しかし、これだけでは右隣のセルの場所が判明しただけです。参照させる為には、まずは それぞれ求めた行番号と列番号を組み合わせてセル番地にしなければなりません。具体的 に言うと「C2」にする必要があります。その為に使用されているのが次にご紹介する関数 です。

ADDRESS 関数
=ADDRESS(行番号,列番号)
指定された <i>行番号と列番号</i> を組み合わせてセル参照形式の文字列を返
र्च
例)=ADDRESS(2,3)
計算結果: \$C\$2

計算結果を見ると分かる通り、列も行も固定するという意味の「\$」が付いていますが、 今回の用途と作成している数式を考えると付いていても付いていなくても変わりがない為、 このまま利用することにします。

それよりもここで問題にするべきことは ADDRESS 関数の「文字列を返す」という部分 です。実際に ADDRESS 関数のみを下図のように書いてみると分かりますがセル番地が表 示されるだけです。

C2	- = X	- 1	/ 1	\hat{x}	=4	١DD	RE	SS(RO	W()	,00	DLU	ΜN	())
A	B C	D	E F	G	Н	Ι	J	К	L	М	N	0	Ρ	Q
1														
2	\$C\$2													

こうなる理由は「文字列を返す」という言葉の通り、本当に文字列としてこのセル番地は 作成されている為です。しかしこのセルを参照したいが為にセル番地を作成しているので す。従って、このセル番地を参照させる為に次の関数を使用します。

INDIRECT 関数 =INDIRECT (*参照文字列*) 指定された*参照文字列*への参照を返す

これにより、右隣のセルを参照することが出来るようになりました。尚、概要を見て頂け れば分かる通り、INDIRECT 関数と ADDRESS 関数は組み合わせて使うことが多いので、 一緒に覚えて頂ければ幸いです。

さて、長々と記載してきましたがわざわざ右隣のセルを自動的に参照できるようにした 理由は COUNTIF 関数でどのセルを基準にするかだけを変えることで全てのチェックボッ クスを自動化するのが容易になる為です。もしかしたら参照する選択肢のセルが右隣でな い場合もあるかもしれませんが、その場合は ROW 関数もしくは COLUMN 関数に足し引 きする数値を変えて頂ければ幸いです。

5. 特定のデータ加工方法

ここでは限定的な項目の限定的なデータ加工方法についてご紹介してまいります。具体 的な説明となるので応用は出来ない可能性がございます。必要な個所のみご覧下さいませ。

5.1. 郵便番号の先頭に「〒」を付ける方法

MG上で『郵便番号』という項目の入力欄はテキストです。そこに入力可能な形式は以下の2パターンです。

郵便番号	680-0525
都道府県	鳥取県
住所	八頭郡八頭町横田

【パターンA】ハイフンあり形式(推奨形式)

郵便番号	6800525
都道府県	鳥取県
住所	八頭郡八頭町横田

【パターン B】 ハイフンなし形式

帳票設定でこれを出力する設定をすると、上図の通りに出力されます。しかしながら、郵 便番号なのだから先頭に「〒」が付いていて欲しいという場合に行う方法をここではご紹介 致します。

今回は表示形式を使用致します。従って、まずは「<u>3.1. 表示形式</u>」にて手順をご確認頂いた上で以降をご覧頂けますようお願い申し上げます。



【パターンA】ハイフンあり形式(推奨形式)

A1	• : 🗙 🖌 ƒ 123	34567
1	〒123-4567	①人力された文字列 ? ×
2	「東京形式」配置	フォント 罫線 塗りつぶし 保護
3	 ③結果的に表示 	
4	された文字列 (<u>氧(C</u>):	-
5	~ ~ ~ ~ ~ ~ ~ ~ ~ ~ ~ ~ ~ ~ ~ ~ ~ ~ ~	, サンプルー
6		〒123-4567
7		
8	三日付	
9		"〒"000-0000
10	パーセンテージ	h:mm
11	分数	h:mm:ss ②設定した書式
12	日奴 文字列	h"時"mm"分 b"咭"mm"分"cc"秒"
13	その他	vvvv/m/d h:mm
14	ユーザー定義	mm:ss
15		mm:ss.0
10		() [h]mmuss
10		[[1]:1111:55 [<=9999999]"∓"000-0000:"∓"@
10		" 〒 "000-0000 ♥
20		
20	×	削除(<u>D</u>)
- 21		

【パターン B】 ハイフンなし形式

さて、早速答えを書きましたが、どちらも先頭に「"〒"」と記載しております。これは「〒」 という文字列を固定文字としてそのまま出力させる為に今回の目的からして必須です。そ の後の設定書式がパターンAとパターンBで変わっています。パターンAの「@」は文字 列を示します。パターンBは7桁以下の数値がそこに入ってくることを前提に前3桁と後 ろ4桁の間に「・」を挿入した文字列にするというものです。つまり、入力された文字列は 7桁の数値でなければ想定通りの結果は得られないでしょう。

しかしながら、『郵便番号』にパターンAとパターンBの両方の書き方をしているという 場合もあるでしょう。その場合は、Excel にそれを勝手に察知してと願うのではなく、そう して貰えるような設定書式にする必要があります。



先程のパターンAの設定書式「"〒"@」とパターンBの設定書式「"〒"000-0000」が入っ ているのは分かると思います。その他に記載されている主な内容は「[<=9999999]」です。 これはどちらのパターンを使うのかを判定させる為の条件になります。入力文字列が 「9999999」以下、つまり7桁の数値以下であるのならパターンBの設定書式を、そうで なければパターンAの設定書式を使って文字列を表示するというものです。

条件である「[<=9999999]」のすぐ後にこの条件に当て嵌まった時に使う設定書式を記述 し、その後にその条件に当て嵌まらなかった時に使う設定書式を記述します。その境目に「;」 と記述することでそこまでが一つの設定書式という目印となります。この特性を使い、分岐 を更に増やし、あらゆるパターンに対応させることも可能ですので、ご参考にして頂ければ 幸いです。

以上

索引

ADDRESS 関数	
COLUMN 関数	
COUNTIF 関数	
FIND 関数	
IF 関数	
INDIRECT 関数	
LEFT 関数	
LEN 関数	
RIGHT 関数	
ROW 関数	
SUM 関数	